
Have It Your Way…
… in Two Steps

1. Reorder the variables in the dataset
(a) RETAIN statement in DATA step, or
(b) SELECT statement in PROC SQL.

2. Export the dataset to Excel:
PROC EXPORT

proc export data=towns(keep=community popn_change popn_2000 popn_1990 county) 
  outfile = "C:\BASUG_demo\towns1.xls"  
  dbms=Excel replace; 
  sheet="Population Change";  * Sheet= is for SAS 9, not recognized by SAS 8; 
run; 

What won’t work …

SAS Usage Note #16111:

In order to reorder the variables/columns when exporting SAS data to 
Excel, you will need to change the order in the SAS file first and export 
the reordered data.

You can not perform the reorder in the PROC EXPORT code.
Refer to #8395, "How to reorder variables in a SAS data set"

Futile attempt: KEEP statement does not reorder variables!

                                                            Per_ 
                          Popn_  Popn_            Income_  Pupil_   Popn_ Popn_ 
Obs  County   Community   1990   2000  Sq_Miles per_Capita   Exp    Sq_Mi Change 
 
  1 Suffolk   Boston     574283 589141   48.4      15581     8880  12,172   2.6 
  2 Norfolk   Brookline   54718  57107    6.8      29044     9002   8,398   4.4 
  3 Norfolk   Braintree   33836  33828   13.9      18624     6882   2,434  -0.0 
  4 Norfolk   Canton      18530  20755   18.9      22035     6956   1,098  12.0 
  5 Norfolk   Wellesley   26615  26613   10.2      32253     8563   2,609  -0.0 
  6 Middlesex Cambridge   95802 101355    6.4      19879    12625  15,837   5.8 
  7 Middlesex Arlington   44630  42389    5.2      21449     7571   8,152  -5.0 
  8 Middlesex Newton      82585  83829   18.1      28840     8978   4,631   1.5 
  9 Middlesex Belmont     24720  24194    4.7      26793     6938   5,148  -2.1 
 10 Norfolk   Dedham      23782  23464   10.5      19045     8034   2,235  -1.3 
 11 Norfolk   Norwood     28700  28587   10.5      18193     7202   2,723  -0.4 
 12 Norfolk   Quincy      84985  88025   16.8      17436     6823   5,240   3.6 
 13 Norfolk   Wrentham     9006  10554   22.2      15856     6127     475  17.2 
 14 Norfolk   Foxboro     14637  16246   20.1      18329     6370     808  11.0 
 15 Norfolk   Walpole     20212  22824   20.5      20128     6128   1,113  12.9 
 16 Middlesex Somerville  76210  77478    4.1      15179     8359  18,897   1.7 
 17 Middlesex Framingham  64989  66910   25.1      20407     7086   2,666   3.0 
 18 Middlesex Hopkinton    9191  13346   26.6      22497     5810     502  45.2 

TOWNS Dataset “in order of appearance”
(i.e. variable creation order)

Futile result:

The KEEP statement 
does not  reorder 
variables.

The variable order 
in the SAS dataset 
determines the Excel 
column order.

What will work… (Method A)

1. Reorder the variables in the dataset
(a) RETAIN statement in DATA step, or
(b) SELECT statement in PROC SQL.

*----------------------------------------------*; 
*----- DATA step, RETAIN statement method -----*; 
 
data pop_change; 
  retain Community Popn_Change Popn_2000 Popn_1990 ;
  set towns; 
run; 
proc sort data=pop_change; 
  by county descending popn_change; 
run; 

What will work… (Method B)

1. Reorder the variables in the dataset
(a) RETAIN statement in DATA step, or
(b) SELECT statement in PROC SQL.

*---------------------------------------------*; 
*----- Proc SQL, SELECT statement method -----*; 
 
proc sql; 
  create table pop_change as 
  select community, popn_change, popn_2000, popn_1990, * 
  from work.towns 
  order by county, popn_change desc ; 
quit; 

        The CONTENTS Procedure 
 
WORK.TOWNS           Observations  18 
DATA                 Variables      9 
 
     Variables in Creation Order 
 
#    Variable             Type    Len 
 
1    County               Char      9 
2    Community            Char     10 
3    Popn_1990            Num       8 
4    Popn_2000            Num       8 
5    Sq_Miles             Num       8 
6    Income_per_Capita    Num       8 
7    Per_Pupil_Exp        Num       8 
8    Popn_Sq_Mi           Num       8 
9    Popn_Change          Num       8 

        The CONTENTS Procedure 
 
WORK.POP_CHANGE      Observations  18 
DATA                 Variables      9 
 
     Variables in Creation Order 
 
#    Variable             Type    Len 
 
1    Community            Char     10 
2    Popn_Change          Num       8 
3    Popn_2000            Num       8 
4    Popn_1990            Num       8 
5    County               Char      9 
6    Sq_Miles             Num       8 
7    Income_per_Capita    Num       8 
8    Per_Pupil_Exp        Num       8 
9    Popn_Sq_Mi           Num       8 

Variable Order has changed!
Original TOWNS Dataset POP_CHANGE Dataset 

after DATA step RETAIN statement,
or SELECT statement in PROC SQL

RETAIN 
stmt.

SAS Usage Note #8395: 
How to reorder variables in a SAS data set

Any of the following statements can be used to change 
the order of variables in a SAS data set:  

ATTRIB, ARRAY, FORMAT, INFORMAT, LENGTH, 
and RETAIN.

FYI: Other Step 1 methods…

For details, read the entire Note #8395.

http://support.sas.com/

Click on “Samples & SAS Notes”. 
Enter “8395” in Search box.

What will work… EXPORT step

2. Export the dataset to Excel:
PROC EXPORT

*-----------------------------------------------*; 
*----- SAS 9 Code for Proc Export to Excel -----*; 
 
proc export data=pop_change(keep=county community popn_1990 popn_2000 popn_change) 
  outfile="C:\BASUG_demo\towns2.xls"  
  dbms=Excel replace; 
*  sheet="Population Change"; 
run; 

*-----------------------------------------------*; 
*----- SAS 8 Code for Proc Export to Excel -----*; 
 
proc export data=pop_change(keep=county community popn_1990 popn_2000 popn_change) 
  outfile="C:\BASUG_demo\towns2.xls"  
  dbms=Excel replace; 
run; 

Sheet=“sheetname” statement became available in SAS 9.

Successful result:

Column order 
(i.e. variable order) 
matches the RETAIN  
or SELECT statement.

The default Excel 
Sheet name is the 
SAS dataset name.  

(In this case, 
POP_CHANGE.)

Creating Multiple Sheets 
in One Excel Workbook

• Must use SAS 9
• Use a PROC EXPORT for each Sheet.
• Use Sheet=“Sheetname” statement.
• Can add a Sheet to an existing Excel file, 

or create a new Excel file.
• dbms=Excel replace;

Can overwrite a Sheet by the same name.
Can not replace an existing Excel file.

*-------------------------------------------------*; 
*----- Multiple Sheets in One EXCEL Workbook -----*; 
*-----              SAS 9 only               -----*; 
 
proc sql; 
  create table pop_density as 
  select county, community, popn_sq_mi, popn_2000, sq_miles 
  from work.towns 
  order by county, popn_sq_mi descending ; 
quit; 
proc export data=pop_density 
  outfile = "C:\BASUG_demo\towns2.xls"  
  dbms=Excel replace; 
  sheet="Population Density 2000"; 
run; 
 
proc sql; 
  create table educ_spend as 
  select community, per_pupil_exp, income_per_capita  
  from work.towns 
  order by per_pupil_exp; 
quit; 
proc export data=educ_spend 
  outfile = "C:\BASUG_demo\towns2.xls"  
  dbms=Excel replace; 
  sheet="Education Spending - 2000"; 
run; 

In SQL, it’s officially 
desc – but Proc SQL in 
SAS also recognizes 
descending

Success:

Sheets were 
added to 

existing file 
(Towns2.XLS)

Sheet Name 
Nuances:

Note that 
SAS replaces 

spaces and 
some 

characters 
with “_”

Success (continued):

Same file 
(Towns2.XLS)

Showing the 
Population Density 

sheet

*-------------------------------------------------*; 
*----- Overwrite an Existing EXCEL Worksheet -----*; 
*-----              SAS 9 only               -----*; 
 
proc sql; 
  create table popn_density as 
  select community, popn_2000, sq_miles, popn_sq_mi  
  from work.towns 
  order by popn_sq_mi; 
quit; 
proc export data=popn_density 
  outfile = "C:\BASUG_demo\towns2.xls"  
  dbms=Excel replace; 
  sheet="Population Density 2000"; 
run; 

Overwrite an Existing Excel Sheet
Let’s re-do the “Population Density 2000” sheet, 

using the REPLACE option.

This time ignore County and sort in ascending order.

Replaced:

This new 
version of 

the sheet is 
sorted by 

Population per 
Square Mile. 

County is 
now omitted.

What if I’m Running SAS 8 ?

• Only one Sheet per Excel file.
Sorry, you’ll have to deal with more files!

• Sheet name will automatically be the 
SAS dataset name.

• Use a PROC EXPORT for each dataset
Specify a different Outfile=“Filename” for each.

• dbms=Excel replace;
Overwrites an existing Excel file if the Outfile name
already exists.

SUMMARY
You can have it your way… in two steps!

1. Reorder the variables in the dataset
(a) RETAIN statement in DATA step, or
(b) SELECT statement in PROC SQL.

2. Export the dataset to Excel:
PROC EXPORT

Advantage of PROC SQL: 
you can sort without a PROC SORT!

Thanks to Vanessa Hayden for the 
PROC SQL variable reordering tip 
that led to this presentation.

You never know what new tip you’ll pick up 
at a BASUG meeting!

Mark A. Martin, Siemens Healthcare Diagnostics Inc.
mark.martin@siemens.com

